Modern Assembly Language Programming with the ARM processor

Chapter 3: Load/Store and Branch Instructions

1. Introduction
2. Load/Store Instructions
3. Branch Instructions

ARM User Program Registers



- Thirteen general-purpose registers (`r0-r12`)
- The stack pointer (`r13` or `sp`)
- The link register (`r14` or `lr`)
- The program counter (`r15` or `pc`)
- Current Program Status Register (`CPSR`)

Hardware-Related Register Rules

- All instructions can access `r0`–`r14` directly.
- Most instructions also allow use of the program counter (`r15`).
- Specific instructions to allow access to `CPSR`.
- `r14`, `r15`, and `CPSR` are “hardware special”.

CPSR



Negative: This bit is set to one if the signed result of an operation is negative, and set to zero if the result is positive or zero.

Zero: This bit is set to one if the result of an operation is zero, and set to zero if the result is non-zero.

Carry: This bit is set to one if an add operation results in a carry out of the most significant bit, or if a subtract operation results in a borrow. For shift operations, this flag is set to the last bit shifted out by the shifter.

oVerflow: For addition and subtraction, this flag is set if a signed overflow occurred.

Conditional Execution

```
op{<cond>}  operands
```

<cond>	English meaning
al	always (this is the default <cond>
eq	Z set (=)
ne	Z clear (≠)
ge	N set and V set, or N clear and V clear (≥)
lt	N set and V clear, or N clear and V set (<)
gt	Z clear, and either N set and V set, or N clear and V set (>)
le	Z set, or N set and V clear,or N clear and V set (≤)
hi	C set and Z clear (unsigned >)
ls	C clear or Z (unsigned ≤)
hs	C set (unsigned ≥)
cs	Alternate name for HS
lo	C clear (unsigned <)
cc	Alternate name for LO
mi	N set (result < 0)
pl	N clear (result ≥ 0)
vs	V set (overflow)
vc	V clear (no overflow)

Instruction Categories

- Load/Store Instructions
- Branch Instructions
  - Branch with Link (subroutine call)
  - Conditional Branches
- Data processing Instructions
  - Arithmetic Operations
  - Logical Operations
  - Comparison Operations
  - Data Movement Operations
  - Multiplication Operations
- Special Instructions
- Pseudo-Instructions

Simplified Computer



Pointers and Addresses

Data must be copied to a register before it can be used in any calculation, but there are not many registers.

In assembly, *almost all* data is accessed using its address in memory.

1. Every memory location has an address.
2. A pointer is a variable that holds an address.
3. A pointer can be stored in a register (short term) or in memory (long term).
4. Before it can be used to access the data it points to, a pointer variable must be loaded into a register.

The address of a statically allocated variable, `x`, can be loading using the following pseudo-instruction:

```
ldr r4, =x
```

This creates a temporary pointer variable in register `r4`, which can then be used to load data from variable `x`.

Addressing Modes

Most of the Load/Store instructions use an <address> which is one of the ten options listed below.

Syntax	Name
[Rn]	Register immediate
[Rn, #± <offset_12>]	Immediate offset
[Rn, ±Rm]	Register offset
[Rn, ±Rm, <shift> #<shift_imm>]	Scaled register offset
[Rn, #±<offset_12>]!	Immediate pre-indexed
[Rn, ±Rm]!	Register pre-indexed
[Rn, ±Rm, <shift> #<shift_imm>]!	Scaled register pre-indexed
[Rn], #±<offset_12>	Immediate post-indexed
[Rn], ±Rm	Register post-indexed
[Rn], ±Rm, <shift> #<shift_imm>	Scaled register post-indexed

<shift> can be any of the shift or rotate operations that will be covered later.

[Rn] is just shorthand notation for [Rn, #0]

Load/Store Operations

- Operations:

LDR / STR	Load/Store 32 bits
LDRH / STRH	Load/Store 16 bits unsigned
LDRB / STRB	Load/Store 8 bits unsigned
LDRSH	Load 16 bits signed
LDRSB	Load 8 bits signed

- Syntax:

```
    <Opcode>{<cond>}{<size>}   Rd, <address>
```

- Examples

```
1  ldrsh    r5, [r2]         @ Load r5 with signed
2                             @ half-word at the address in r2
3  strb     r1, [r9, #4]     @ Store the byte in r1 at
4                             @ the address (r9 + 4)
5  ldr      r5, [r3, r2]!    @ Load r5 with word at the
6                             @ address (r3 + r2), then
7                             @ store the address in r3
```

Load/Store Examples

```
1  ldrh    r9, [r2, #2]!  @ Load r9 with halfword at the
2                          @ address (r2 + 2), then store
3                          @ the address in r2
4  ldrsh   r5, [r2]       @ Load r5 with signed
5                          @ half-word at the address in r2
6  strb    r1, [r9, #4]   @ Store the byte in r1 at
7                          @ the address (r9 + 4)
8  ldr     r7, [r3, r2]!  @ Load r7 with word at the
9                          @ address (r3 + r2), then
10                         @ store the address in r3
11 ldrh    r9, [r2, #2]!  @ Load r9 with halfword at the
12                         @ address (r2 + 2), then store
13                         @ the address in r2
14 ldr     r7, [r3], #4   @ Load r7 with word at the
15                         @ address in r3 then increment
16                         @ r3 by 4
```

Load/Store Multiple Registers

These instructions are used to store registers on the stack, and for copying blocks of data. There are four variants for the LDM and STM instructions, and each variant has two equivalent names.

- Operations:
  LDM / STM Load/Store Multiple Registers
- Syntax:
  LDM|STM{<variant>} Rd{!}, {<list>}^
- The trailing ^ can only be used by operating system code.
- <variant> is chosen from the following table:

Block Copy		Stack	
IA	Increment After	EA	Empty Ascending
IB	Increment Before	FA	Full Ascending
DA	Decrement After	ED	Empty Descending
DB	Decrement Before	FD	Full Descending

Load/Store Multiple Registers (Continued)

The C compiler always uses the `stmfd` and `ldmfd` versions for the stack.

- Examples

```
1  stmfd sp!,{r4-r7,fp,lr}@ store r4, r5, r6, r7, r11, and
2                         @ r14 on the stack, and store the
3                         @ new stack pointer in sp
4  ldmfd sp!,{r4-r7,fp,lr}@ load r4, r5, r6, r7, r11, and
5                         @ r14 from the stack, and store
6                         @ the new stack pointer in sp
7  stmib r9!,{r0-r7}      @ Store 8 registers at the
8                         @ location pointed to by r9, and
9                         @ increment r9 BEFORE each store.
10                        @ After executing, r9 will point
11                        @ to the last item stored.
12 ldmia r4,{r0,r2,r3}    @ Load r0, r2, and r3 at the
13                        @ location pointed to by r4 and
14                        @ increment the address AFTER
15                        @ each store. After executing, r4
16                        @ will contain its original value.
```

Block Copy Example

```
1          .data
2  source: .word  12
3          .word  23
4          .word  43
5          .word  33
6          .word  12
7          .word  23
8          .word  6
9          .word  13
10 dest:   .skip  32
```

```
1          stmfd   sp!{r0-r9} @ push r0...r9 to the stack
2          ldr     r8,=source @ load address of source
3          ldr     r9,=dest   @ load address of destination
4          ldmia   r8,{r0-r7} @ load eight words from source
5          stmia   r9,{r0-r7} @ store them in destination
6          ldmfd   sp!{r0-r9} @ restore contents of r0...r9
```

Atomic Load-Store

Multiprogramming and threading require the ability to set and test values *atomically*. These instructions are used by the Operating System and/or threading libraries to guarantee *mutual exclusion*.

- Operations:

  SWP Load a word and store a word in one atomic operation.

  SWPB Load a byte and store a byte in one atomic operation.

- Syntax:
  SWP{<cond>}{B} Rd, Rm, [Rn]

- Example

```
1 swpeqb  r1, r4, [r3]  @ if (EQ) then load r1 with byte
2                       @ at address in r3 and store byte
3                       @ in r4 at address in r3
```

Note: SWP and SWPB are deprecated in favor of LDREX and STREX.

Mutual Exclusion - New Method

Exclusive load (`ldrex`) reads data from memory, tagging the memory address at the same time. Exclusive store (`strex`) stores data to memory, but only if the tag is still valid. Every memory access to the same address between `ldrex` and `strex` will invalidate the tag. This provides mutual exclusion on multiprocessor systems.

- Operations:

  LDREX Load register from memory and tag the memory address

  STREX Store register in memory if tag is valid, and report success

- Syntax:
  LDREX|STREX{<cond>}{{S}<size>} Rd, <address>

- Example

```
1       ldr     r12, =sem       @ preload semaphore address
2       ldr     r1, =LOCKED     @ preload "locked" value
3 spin_lock:
4       ldrex   r0, [r12]       @ load semaphore value
5       cmp     r0, r1          @ if semaphore is not locked
6       strexne r0, r1, [r12]   @    try to claim
7       cmpne   r0, #1          @    and check success
8       beq     spin_lock       @ retry if claiming failed or
9                               @ it was already locked.
```

Branch Instructions

- Operations:
  - B load pc with new address (branch)
  - BL Save pc in lr, then load pc with new address (branch and link)
- Syntax:
  B{L}{<cond>} <target_address>
- Example

```
1      mov  r4, #10     @ load 10 into r4
2 loop_a:
3      mov  r0, #1      @ fd -> stdout
4      ldr  r1, =msg    @ buf -> msg
5      ldr  r2, =len    @ count -> len(msg)
6      mov  r7, #4      @ write is syscall #4
7      swi  #0          @ invoke syscall
8      subs r4, #1      @ decrement loop counter
9      bne  loop_a      @ repeat until loop counter is zero
```